

EYE COORDINATES

PURPOSE

Specifies the eye coordinates (i.e., the viewing position) for subsequent 3-dimensional plotting (via the 3D-PLOT command).

DESCRIPTION

Changing the viewing position allows you to view a 3D plot from a different perspective. This feature can sometimes allow you to see more clearly certain structures that are hidden in the default viewing position.

SYNTAX

EYE COORDINATES <x> <y> <z>

where <x> is a decimal number or parameter that specifies the eye coordinate for the X axis;

<y> is a decimal number or parameter that specifies the eye coordinate for the Y axis;

and <z> is a decimal number or parameter that specifies the eye coordinate for the Z axis.

Entering a "." for one of the coordinates specifies that the previous value is used.

EXAMPLES

EYE COORDINATES 20 20 3.5

EYE COORDINATES 10 . .

NOTE 1

If the 3D plot is compressed in one or more directions, the most likely problem is that the X, Y, and Z scales have different ranges (e.g., X and Y go from 0 to 1000 while Z goes from 0 to 1). One solution to this problem is to scale the data to the same range via the LET command (e.g., divide each of them by the appropriate power of 10 so that they all go from -1 to +1). Changing the eye coordinates is an ineffective way to compensate for unequal scales.

NOTE 2

Increasing the magnitude of the eye coordinates shrinks the size of the 3D plot. That is, the further away the eye is from the plot, the smaller the plot appears. However, if the magnitude is too small, the plot may be clipped.

NOTE 3

The eye coordinates can be negative. This can be useful for looking at the plot from a different perspective. The sample program demonstrates the approach of showing the positive view, the negative X view, the negative Y view, and the X and Y both negative view on a single page. Most reasonable views generate plots that are only marginally different from one of these 4 views. Changing the magnitude of the eye coordinates can make the plot slightly larger or smaller, but will not change the basic appearance. Making the Z eye coordinate negative is generally not helpful.

DEFAULT

default x = data xmax + 3 x (data xmax - data xmin)

default y = data ymax + 3 x (data ymax - data ymin)

default z = data zmax + 3 x (data zmax - data zmin)

SYNONYMS

EYE is a synonym for EYE COORDINATE

RELATED COMMANDS

3D-PLOT	=	Generates a 3-d data or function plot.
ROTATE EYE	=	Automatically rotate the eye coordinates by a specified number of degrees.
3D FRAME	=	Specify the type of frame drawn for a 3-d plot.

APPLICATIONS

3D plotting

IMPLEMENTATION DATE

Pre-1987

PROGRAM

```

LET FUNCTION F1 = X**2 + Y**2 - X*Y
LET X = SEQUENCE -2 .2 2;LET Y = SEQUENCE -2 .2 2; LET Z = F1
MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
LET XMAX = MAXIMUM X; LET XMIN = MINIMUM X; LET XEYE = XMAX + 3*(XMAX-XMIN)
LET YMAX = MAXIMUM Y; LET YMIN = MINIMUM Y; LET YEYE = YMAX + 3*(YMAX-YMIN)
LET ZMAX = MAXIMUM Z; LET ZMIN = MINIMUM Z. LET ZEYE = ZMAX + 3*(ZMAX-ZMIN)
. All views positive
EYE COORDINATES XEYE YEYE ZEYE
TITLE ALL EYE COORDINATES POSITIVE
3D-PLOT F1 FOR X = -2 .2 2 FOR Y = -2 .2 2
. X view negative
LET XTEMP = -XEYE
EYE COORDINATES XTEMP YEYE ZEYE
TITLE X VIEW NEGATIVE
3D-PLOT F1 FOR X = -2 .2 2 FOR Y = -2 .2 2
. Y view negative
LET YTEMP = -YEYE
EYE COORDINATES XEYE YTEMP ZEYE
TITLE Y VIEW NEGATIVE
3D-PLOT F1 FOR X = -2 .2 2 FOR Y = -2 .2 2
. Both X and Y views negative
EYE COORDINATES XTEMP YTEMP ZEYE
TITLE X AND Y VIEWS NEGATIVE
3D-PLOT F1 FOR X = -2 .2 2 FOR Y = -2 .2 2
END OF MULTIPLY

```

